
Далее… 

• Функции 
– Рекурсия 
– Выравнивание стека 
– Различные соглашения о вызове функций 

• cdecl/stdcall/fastcall, отказ от указателя фрейма 
• Соглашение вызова для x86-64 

– Переменное число параметров 
– Вызов по указателю 
– Ассемблерные вставки 
– Переполнение буфера, эксплуатация ошибок,  

механизмы защиты  

• Операции над строками / Строковые команды  
• Динамическая память 
• Числа с плавающей точкой 

«Заглянуть за горизонт» 

void f(int i) { 
   int a[3] = {1, 2, 3}; 
   printf("%x\n", a[i]); 
} 

f: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 40 
   mov     eax, dword [ebp+8] 
   mov     dword [ebp-20], 1 
   mov     dword [ebp-16], 2 
   mov     dword [ebp-12], 3 
   mov     eax, dword [ebp-20+eax*4] 
   mov     dword [esp], .LC0 
   mov     dword [esp+4], eax 
   call    printf 
   leave 
   ret 

i Вывод на 
экран 

0 1 

1 2 

2 3 

3 b7ff1030 

4 8049ff4 

5 bffff748 

6 804844b 

7 7 

8 b7fccff4 

9 8048470 
3 

содержимое фрейма 
вызывающей функции 

массив a 

«мусор» 

сохраненный ebp 

адрес возврата 

фактический аргумент 

Переполнение буфера 

void function(char *str) { 
   char buffer[16]; 
   strcpy(buffer,str); 
} 
 
int main(void) { 
  char large_string[256]; 
  int i; 
 
  for( i = 0; i < 255; i++) 
    large_string[i] = 'A'; 
 
  function(large_string); 
} 

function: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 40 
   mov     eax, dword [ebp+8] 
   mov     dword [esp+4], eax 
   lea     eax, [ebp-24] 
   mov     dword [esp], eax 
   call    strcpy 
   leave 
   ret 

Немного поменяем содержимое переменной large_string 
“AAA…AA\xB0\x92\x04\x08AAAA\x38\xF7\xFF\xBF/bin/sh\x00” 

Пусть в момент вызова адрес 
возврата был размещен в стеке 
по адресу 0xbffff72с, адрес 
функции system – 0x80492b0 

28 байт с 
символом ‘A’ 

#include <stdlib.h>  
int system(const char *command);  

http://linux.die.net/include/stdlib.h

Переполнение буфера 

void function(char *str) { 
   char buffer[16]; 
   strcpy(buffer,str); 
} 
 
int main(void) { 
  char large_string[256]; 
  int i; 
 
  for( i = 0; i < 255; i++) 
    large_string[i] = 'A'; 
 
  function(large_string); 
} 

function: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 40 
   mov     eax, dword [ebp+8] 
   mov     dword [esp+4], eax 
   lea     eax, [ebp-24] 
   mov     dword [esp], eax 
   call    strcpy 
   leave 
   ret 

Содержимое переменной large_string – эксплоит, т.е. входные 
данные, приводящие к эксплуатации уязвимости (ошибки)  
“AAA…AA\xB0\x92\x04\x08AAAA\x38\xF7\xFF\xBF/bin/sh\x00” 

Ошибка переполнения буфера 
порождает уязвимость 

атака 
return-to-libc  

Способы защиты: 
Канарейка на стеке 

#include <stdio.h> 
#include <string.h> 
 
int my_bad_function(char* d_msg) 
{ 
    char what[100]; 
    strcpy(what, d_msg); 
    printf(“%s\n”, what); 
} 

В последних версиях 
компилятора gcc проверка 
включена по-умолчанию.  
Отключается опцией  
-fno-stack-protector 

   ...             ; пролог 
   mov     eax, dword [gs:20] 
   mov     dword [ebp-12], eax 
   xor     eax, eax 
   ...             ; тело функции 
   mov     edx, dword [ebp-12] 
   xor     edx, dword [gs:20] 
   je      .L3 
   call    __stack_chk_fail 
.L3: 
   ...             ; эпилог 

Способы защиты: 
… и не только  

• Неисполняемый стек и данные 
 

 
 
 
 
 
 
 

• Address Space Layout Randomization (ASLR) 
• Безопасные библиотеки  

errno_t strcpy_s( char *strDestination,  
size_t numberOfElements,  
const char *strSource );  

char buffer[] = {...};  
 
typedef void (* func)(void); 
 
int main(int argc, char** argv) { 
   func f = (func) buffer; 
   f(); 
   return 0; 
}  

7 

_chk-версии некоторых стандартных функций 

void f(int i) { 
   int a[3] = {1, 2, 3}; 
   printf("%x\n", a[i]); 
} 

f: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 40 
   mov     eax, dword [ebp+8] 
   mov     dword [ebp-20], 1 
   mov     dword [ebp-16], 2 
   mov     dword [ebp-12], 3 
   mov     eax, dword [ebp-20+eax*4] 
   mov     dword [esp+4], .LC0 
   mov     dword [esp], 1 
   mov     dword [esp+8], eax 
   call    __printf_chk 
   leave 
   ret 

i Вывод на экран 

0 1 

1 2 

2 3 

3 b7702030 

4 8049ff4 

5 bfbcb3b8 

6 804846b 

7 7 

8 b781aff4 

9 8048490 
8 

GCC FORTIFY_SOURCE  

void foo(char *string) { 
   char buf[20]; 
   strcpy(buf, string); 
} 

foo: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 56 
   mov     eax, dword [ebp+8] 
   mov     dword [esp+4], eax 
   lea     eax, [ebp-28] 
   mov     dword [esp], eax 
   call    strcpy 
   leave 
   ret 9 

foo: 
   push    ebp 
   mov     ebp, esp 
   sub     esp, 56 
   mov     eax, dword [gs:20] 
   mov     dword [ebp-12], eax 
   xor     eax, eax 
   mov     eax, dword [ebp+8] 
   mov     dword [esp+8], 20 
   mov     dword [esp+4], eax 
   lea     eax, [ebp-32] 
   mov     dword [esp], eax 
   call    __strcpy_chk 
   mov     eax, dword [ebp-12] 
   xor     eax, dword [gs:20] 
   jne     .L5 
   leave 
   ret 
.L5: 
   call    __stack_chk_fail 

Отключается макросом  
-D_FORTIFY_SOURCE=0 

В последних версиях компилятора 
gcc по-умолчанию включается 
вместе с оптимизацией.  

gcc –v 
... 
gcc version 4.4.3 

stack-protector 
+ 
FORTIFY_SOURCE 

Неисполняемый стек и данные / ASLR / … 

Операционная 
система  

ASLR 

Стек и куча Библиотеки Образ 
программы 

Ubuntu 10.04 Да Да Да Зависит от 
сборки прог. 

Debian Sarge HW Да Да Зависит от 
сборки прог. 

Windows Vista, 7 HW Да Зависит от 
сборки прог. 

Зависит от 
сборки прог. 

Mac OS X 10.6 HW Нет Да Нет 

XW

Некоторые программы 
требуют возможность 
размещать и 
выполнять код на стеке 
(just-in-time 
компиляция) 

Программа защищается комплексом различных средств 
«Ответ» атакующей стороны: 
• ROP – return oriented programming 
• Инструменты статического и динамического анализа 

• Фаззинг и символьная интерпретация  
• и много другое ... 

Используют 
как 
разработчики, 
так и хакеры 

Вызов функций – заключение 

• Порядок вызова функций образует стек (call / ret) 
• Если P вызывает Q, то Q завершается до завершения P 

• Рекурсия (в том числе косвенная ) корректно 
реализуется через общее соглашение о вызове 
функций 
– Фрейм используется для размещения локальных 

переменных и сохранения значений регистров 
– Аргументы для вызова очередной функции размещаются 

на «верхушке» стека 
– Результат возвращается через регистр eax 

• Параметры передаются по значению  
• cdecl – стандартное соглашение для Linux/IA-32 

• Удобно реализовывать функции с 
переменным числом параметров  

• Существуют различные варианты соглашения 
вызова  

• Ошибки переполнения буфера, размещенного на 
стеке, представляют серьезную угрозу 
безопасности программ  

Адрес возврата 

Сохраненные 

регистры и 

локальные 

переменные 

«Пространство» 

параметров 

«Прежний» ebp 

Аргументы 

Фрейм  

вызывающей  

функции 

ebp 

esp 
11 

Далее ...  

• Функции  

• Операции над строками / Строковые команды  

– Префиксы повтора  

• Динамическая память 

• Числа с плавающей точкой 

– Представления для вещественных чисел 

– Сопроцессор x87 

12 

Сравнение строк 
специализированные команды 

• CMPSB сравнение байт 
• CMPSW сравнение 16-разрядных чисел  
• CMPSD сравнение 32-разрядных чисел 
• CLD/STD – очистить/установить флаг DF  

Tmp = [ESI] – [EDI]; 
 
Установить статусные флаги согласно значению TMP; 
 
If (DF == 0) { 
   ESI += sizeof(операнд); 
   EDI += sizeof(операнд); 
} else { // DF == 1 
   ESI -= sizeof(операнд); 
   EDI -= sizeof(операнд); 
} 

Упрощенный псевдокод  из 
документации IA-32/Intel64 



© 2014 МГУ/ВМК/СП 

Сравнение строк равного размера 

CMAIN: 
   push ebp 
   mov  ebp, esp 
   sub  esp, 12 
   mov  dword [esp], s1 
   mov  dword [esp + 4], s2 
   mov  dword [esp + 8], BUFSIZE-1 
   call strange_strncmp ; возвращаем -1,  
                 ; если строки равны, 
                 ; иначе – номер байта, 
                 ; в котором встретилось  
                 ; различие (считаем с 0) 
   PRINT_DEC 4, eax   
   NEWLINE 
   xor  eax, eax 
   mov  esp, ebp 
   pop  ebp 
   ret 
 

%include 'io.inc' 
 
BUFSIZE equ 32 
 
section .data 
 s1 db 'some text' 
    times BUFSIZE-$+s1 db 0 
 
 s2 db 'some text...' 
    times BUFSIZE-$+s2 db 0 
 
section .text 
global CMAIN 

Какое объявление будет у функции 
strange_strncmp? 

Сравнение строк равного размера 

strange_strncmp: 
   push ebp 
   mov  ebp, esp 
   push esi 
   push edi 
   mov  ecx, dword [ebp + 16] 
   mov  esi, dword [ebp + 8] 
   mov  edi, dword [ebp + 12] 
   ; ...  
 
   ; ... 
.end: 
   pop  edi 
   pop  esi 
   mov  esp, ebp 
   pop  ebp 
   ret 

   ; ...  
   mov   eax, -1 
   cld 
   jecxz .end 
 .loop: 
   cmpsb 
   jne   .ne 
   loop  .loop 
   jmp   .end 
 .ne: 
   mov   eax, dword [ebp + 16] 
   sub   eax, ecx 
.end: 
   ; ...  

15 

Формат инструкции 

      CMPSB    A6 

REP   CMPSB  F3 A6 

REPNE CMPSB   F2 A6  16 

Явно записываются в тексте ассемблерной 
программы только префиксы 
REP/REPE/REPNE и LOCK 

Префикс повтора 

• REPE = REPZ = REP // 0xF3 

• REPNE = REPNZ     // 0xF2 

WHILE (ECX != 0) { 
   Выполнить соответствующую строковую инструкцию; 
   ECX = (ECX – 1); // данная операция на флаги не влияет 
   IF (ECX == 0) 
      break; 
   IF (((Префикс повтора == REPZ или REPE) и 
       (ZF == 0)) или  
       ((Префикс повтора == REPNZ) и (ZF == 1) ))) { 
      break; 
   } 
} 

Что будет, если при очередном выполнении 
строковой команды произойдет сбой? 

Интерпретация кода префикса 
определяется кодом команды 
• REP  MOVSD 
• REPZ CMPSB 

Сравнение строк равного размера 

my_strncmp: 
   push ebp 
   mov ebp, esp 
   push esi 
   push edi 
   xor eax, eax 
   mov ecx, dword [ebp + 16] 
   mov esi, dword [ebp + 8] 
   mov edi, dword [ebp + 12] 
   ; ...  
 
   ; ... 
.end: 
   pop edi 
   pop esi 
   mov esp, ebp 
   pop ebp 
   ret 

   ; ...  
   cld 
   repe cmpsb 
   je .end 
   mov eax, 1 
   mov ecx, -1 
   cmovl eax, ecx 
.end: 
   ; ...  

Приводим поведение 
функции к стандартному 
виду 
• меньше    -1 
• равно        0 
• больше      1 

Остальные строковые команды 

• SCAS(B|W|D) 
– Сравниваем аккумулятор со строкой [EDI] 

• MOVS(B|W|D) 
– Копируем строки [EDI]  [ESI] 

• STOS(B|W|D) 
– Выгружаем аккумулятор по адресу [EDI] 

• LODS(B|W|D) 
– Загружаем в аккумулятор из адреса [ESI] 

– Комбинировать с префиксом повтора не 
целесообразно 

19 

memset 

20 

memset: 
   push ebp 
   mov  ebp, esp 
   push edi 
   mov  ecx, dword [ebp + 16] 
   mov  edi, dword [ebp + 8] 
   mov  al,  byte  [ebp + 12] 
   cld 
   rep  stosb 
   pop  edi 
   mov  esp, ebp 
   pop  ebp 
   ret 

#include <string.h> 
 
void *memset(void *s, int c, size_t n); 

Компактные функции 
стандартной библиотеки 
компилятор может встроить 
непосредственно в место вызова  



© 2014 МГУ/ВМК/СП 

strnlen 

strnlen: 
   push ebp 
   mov  ebp, esp 
   push edi 
   xor  eax, eax 
   mov  ecx, dword [ebp + 12] 
   mov  edi, dword [ebp + 8] 
   cld 
   repne scasb 
   mov  eax, dword [ebp + 12] 
   jne  .notNullAtEnd 
   dec  eax 
   sub  eax, ecx 
 .notNullAtEnd: 
   pop  edi 
   mov  esp, ebp 
   pop  ebp 
   ret 

#include <string.h> 
 
size_t strnlen(const char *s, size_t maxlen); 

Если после того, как отработала команда 
«repne scasb», флаг ZF не был 
выставлен, то это означает, что 0 в строке 
так и не нашли и длина строки – maxlen. 


